

Historic, archived document

Do not assume content reflects current scientific knowledge, policies, or practices.

United States
Department of
Agriculture
Economic
Research
Service
National
Economics
Division

Equilibrium Price Solution of Net Trade Models Using Elasticities

Ralph Seeley

U.S. DEPARTMENT OF AGRICULTURE
NATIONAL AGRICULTURAL LIBRARY

SEP 18 1997

CATALOGING PREP.

• Methodically estimating world market equilibrium prices and the effect of policy changes on the world market for major agricultural products is an important task in the field of international trade. This paper presents a technique for doing this which is based on the use of a general equilibrium model. The model is designed to reflect market and government policy effects on agricultural markets. The model is used to estimate world market equilibrium prices and to determine the effect of policy changes on the world market for major agricultural products.

EQUILIBRIUM PRICE SOLUTION OF NET TRADE MODELS USING ELASTICITIES. By Ralph Seeley. International Economics Division, Economic Research Service, U.S. Department of Agriculture. June 1986. ERS Staff Report No. AGES860610.

ABSTRACT

This report shows a solution procedure which can rapidly calculate equilibrium world prices in a large, complex net trade model. The "elasticity solution procedure" works for a model in which price elasticities are fairly stable over time and various prices. The model must accept a world price vector and return a residual world net trade vector. The procedure automatically stays in the region of positive prices and quantities. It builds a complete information set of own- and cross-price elasticities to fit the model's behavior. The procedure runs on the IIASA and GOL world agriculture models with sharp improvements in convergence time over Walrasian tâtonnement and gradient search, and moderate improvements over Newton's method. The procedure can reconcile inconsistent trade estimates made by country analysts.

Keywords: Net trade models, economic model solution, equilibrium prices, implied arc elasticity matrix, negative prices, IIASA world agriculture model, GOL model, simulation, roots of nonlinear equations, numerical analysis

ACKNOWLEDGMENTS

The author thanks Larry Deaton, Clark Edwards, Günther Fischer, M. C. Hallberg, Forrest Holland, Vernon Roningen, and Lloyd Teigen for their review comments. The Food and Agriculture Program of IIASA (International Institute for Applied Systems Analysis) kindly made available a copy of its world agriculture model. Vernon Roningen took part in implementing the elasticity solution procedure in the spreadsheet version of the current GOL (grain, oilseeds, and livestock) model.

CONTENTS

	<u>Page</u>
SUMMARY	iv
INTRODUCTION TO THE ELASTICITY SOLUTION PROCEDURE	1
IMPLIED ARC ELASTICITIES	2
RELATIVE PRICE CHANGES	4
Demand Simultaneous with Prices	4
Supply and Demand Simultaneous with Prices	6
Supply and Demand Simultaneous with Prices but Inversion Limited	7
NEW PRICES	8
RECALCULATION OF ELASTICITIES	9
SIMULATION RESULTS	10
REFERENCES	13
APPENDIX	14
Flowcharts for Elasticity Solution Procedure	14
FORTRAN Listing of Elasticity Solution Procedure for IIASA Model	16

SUMMARY

This report shows a way to find equilibrium world prices in a net trade model. The "elasticity solution procedure" is intended for a model with fairly stable price elasticities over different prices and time. The procedure will operate on a model which can accept a world price vector and return a residual net trade vector. The elasticity procedure requires no additional information; it treats the model as a black box. Although the procedure does not delve into the workings of the model, the procedure does reveal some of the model's aggregate behavior by generating an elasticity matrix. An elasticity matrix can help in making a quick simplified form of the model, in tuning the model, and in examining its stability over time.

The effectiveness of the procedure stems from four main characteristics. The elasticity functions used here naturally stay in the first quadrant of positive prices and quantities. In the calculation of new prices, the procedure avoids any tendency toward negative prices without loss of information. The procedure uses the complete information set of own- and cross-price responses. Simple constant elasticity functions appear to adequately fit the aggregate price responses of the models tested.

The elasticity solution procedure begins by calculating an implied arc elasticity matrix in the first year of endogenous world prices. First, the model is run for 1 year with the previous period's prices or historical prices, giving a base set of quantities. Next, the world price of just the first commodity is changed by a moderate percentage and the model is rerun for the same year. Quantity changes from the base level are measured and used to calculate implied arc elasticities with respect to the shocked price. This approach is repeated for each of the other commodity prices, one at a time.

The procedure can use demand and supply elasticities, or a combination of both. The choice depends mainly on whether the model calculates supply simultaneously with prices. The desired change in net trade is (minus) the current level of net trade. This report derives equations which link the relative price changes to the elasticities and the relative quantity changes.

For a positive price change, the old price increases in proportion to the desired relative price change. For a negative price change, the old price moves a fraction of the way toward zero. Negative prices are thereby prevented. The relationship between the relative price change and the new price has useful mathematical properties. Both the relationship and its first derivative are continuous and positive.

The elasticity solution procedure has been successfully applied to the IIASA (International Institute for Applied Systems Analysis) world agriculture model. The scenario-versus-base comparisons for which the model is used in the Economic Research Service require that world net trade be driven close to zero. When the maximum ratio of net trade to supply for any commodity is set at 0.0001 or 0.00001, the elasticity solution procedure converges an order of magnitude faster on the price-endogenous demand side than do the Walrasian tâtonnement and gradient search procedures used by IIASA. The elasticity procedure converges more reliably and about one and a half times more rapidly than Newton's method. A similar comparison with Walrasian tâtonnement for the GOL (grain, oilseeds, and livestock) model yields a tenfold increase in speed. The elasticity solution procedure can also coordinate forecasts by commodity and country analysts.

Equilibrium Price Solution of Net Trade Models Using Elasticities

Ralph Seeley

INTRODUCTION TO THE ELASTICITY SOLUTION PROCEDURE

Construction of a large economic simulation model is a major undertaking. Equations must be formulated, data collected, and coefficients found. The modeler's work is not done at that point. The model must be solved, tuned, stabilized, documented, and used. This report concentrates on one means of solving a large economic trade model. The solution procedure discussed also provides a first set of information for tuning and stabilization. This procedure should free some modelers from the mechanics of finding equilibrium prices so that their skills will be available for tuning their models and running scenarios.

The solution algorithm outlined in this report will be referred to as the "elasticity solution procedure." It runs on the IIASA (International Institute for Applied Systems Analysis) world agriculture model (3). 1/ The procedure works quite satisfactorily with the May 1984 and March 1985 versions of the model, and with the current version at IIASA. Each version of the model consists of over 30,000 lines of FORTRAN, many of which are reused by a number of commodities and country sub-models. The procedure also works well on the spreadsheet version of the current GOL (grain, oilseeds, and livestock) model (9,11). The elasticity procedure can achieve an order of magnitude improvement in solution time for these models over Walrasian tâtonnement or gradient search (4). It performs more consistently and about one and a half times as rapidly as Newton's method (4). Moreover, the solution procedure is conceptually simple and easy to apply. The procedure may be used to guide a group of commodity and country analysts in coordinating forecasts to find consistent prices and trade.

The elasticity procedure will operate on a trade model which can simply generate a residual net trade vector when presented with a price vector. The solution procedure attempts to rapidly select an equilibrium price vector which drives the net trade vector to zero. At that point, world exports equal imports for each commodity. The elasticity approach works for a model in which own- and cross-price elasticities are fairly stable over time and various prices.

The elasticity solution procedure treats a model as a black box; it examines only the inputs and outputs of the model. The only information used is the

1/ Underscored numbers in parentheses refer to items in the references.

residual net trade vector which the model provides in response to a price vector. The modeler need not spend time reordering equations or calculating derivatives. This feature simplifies the economist's work when the model is quite large and complex, or contains optimizing routines. These factors may reduce the applicability of the Gauss-Seidel (7) approach, and they make minimization of the number of price iterations important. The fixed time needed to estimate the elasticity matrix is most likely to be justified if the model is to be used for multiperiod simulations.

The procedure has the advantage that it makes available the elasticity matrix, which may be used for other purposes. The elasticity matrix provides information useful for stabilizing the model (2,7), for making clear the aggregate responses of the model for tuning, and for generating a quick "reduced form" of the model. In appropriate applications, the elasticity solution procedure enables rapid solution of trade models. The procedure can be discussed in terms of finding roots of systems of nonlinear equations where the independent variables (prices) and some dependent variables (levels of demand and supply) must be positive. That wording comes from the field of numerical analysis. This report uses the narrower economic terminology to make the results more useful to modelers.

This solution procedure has been developed to permit rapid determination of equilibrium prices for the IIASA world agriculture model. The procedure has been conceived of without reference to other sources. A subsequent review of the model solution literature has revealed related work. Although the specifics of this solution procedure have not been discovered in the literature, they may exist in research not found by the author.

The method for calculating the elasticity matrix will be explained first. Next, the report considers use of elasticity matrices to solve three types of models. The equations presented lead to calculation of a relative price change vector. It transforms the price vector from the previous iteration into the new price vector. The report explains how positive prices are ensured. When convergence is too slow, the elasticity matrix is re-estimated. Comparisons of simulation results from other solution procedures follow the equation derivations. Flowcharts and a FORTRAN example of the elasticity solution procedure conclude the discussion.

IMPLIED ARC ELASTICITIES

The solution procedure calculates implied arc elasticities by simply shocking each price in the world price vector, one at a time. The quantity responses for each commodity (changes in demand and supply) are noted and used to calculate elasticities with respect to the price which has been shocked.

Use of shocks to independent variables to examine the responses of systems of equations is mentioned in prior work. Goldfeld and Quandt discuss the numerical evaluation of derivatives as a labor-saving device (4). The first derivatives may be approximated by the slope of the secant over some interval and second derivatives by a change in first derivatives over some interval. Holbrook uses shocks to independent variables to estimate first partial derivatives in a large nonlinear stochastic model (5). Brandsma, Hallett, and van der Windt use shocks to control variables to estimate a matrix of partial derivatives (1). They use this matrix in an optimal control problem. Seeley uses shocks to estimate elasticities in the IIASA world agriculture model (13).

The section below describes estimation of demand elasticities. Supply elasticities may be calculated in the same way. Each of the prices and quantities is measured in the same simulation period. The first calculation of the elasticity matrix occurs in the first period during which prices are endogenous. At that time, the model is run for one period using historical prices or prices from the previous period in the model. This provides a base from which prices and quantities will change so as to enable calculation of the elasticity matrix. Next, the model is rerun for the same year with the world price of commodity_J changed (equation 1). The prices of the other commodities are unchanged. The elasticity is the relative demand change over the relative price change for the price change scenario versus the base (equation 2). The equation defines an own-price elasticity if commodity_i (whose quantity change is measured) equals commodity_J (whose price is changed). When commodity_i differs from commodity_J, the equation represents a cross-price elasticity. Similar scenarios and elasticity calculations for each of the other world prices go together to build the elasticity matrix. The solution procedure may recalculate the elasticities in later years if convergence slows. A section of the report below elaborates on this point.

Symbol definitions appear below. The asterisk (*) denotes multiplication.

i	= Index for commodities or rows.
J	= Commodity whose price is shocked.
D _{b,i}	= Base demand for commodity _i , without price shock.
D _{s,i}	= Scenario demand for commodity _i , with price shock.
P _{b,J}	= Base world price for commodity _J , without price shock.
P _{s,J}	= Scenario world price for commodity _J , with price shock.
ED _{i,J}	= Implied arc elasticity of demand for commodity _i with respect to world price of commodity _J .

$$P_{s,i} = \begin{cases} P_{b,i} * 1.1 & \text{if } i = J \\ P_{b,i} & \text{if } i \neq J \end{cases} \quad \text{for all } i. \quad (1)$$

$$ED_{i,J} = \frac{(D_{s,i} - D_{b,i}) / D_{b,i}}{(P_{s,J} - P_{b,J}) / P_{b,J}} \quad \text{for all } i \text{ and } J. \quad (2)$$

The use of a 10-percent price shock has been empirically justified using the IIASA and GOL models. A 10-percent price shock seems reasonable in estimating each response surface, in part because year-to-year price changes in the IIASA and GOL models are typically on the order of 10 percent. A very small price shock has two disadvantages. First, computer roundoff error becomes more important. Second, a small shock may reduce the ability of the solution procedure to rapidly bridge a small discontinuity in the net trade response surface. A very large price shock also tends to make the estimated arc elasticity deviate too much from the actual arc elasticity corresponding to the desired price change. Examination of the number of iterations required for model solution shows that it is better to err on the side of price shocks that are too large rather than too small.

RELATIVE PRICE CHANGES

The elasticity solution procedure has an advantage over Walrasian tâtonnement in that it uses a more complete set of information. The elasticity procedure takes own- and cross-price effects into account. Cross-price effects can result in a commodity price which should be decreased for convergence even though net imports of the commodity are positive and its own-price elasticity of demand is negative. Walrasian tâtonnement would move this price in the wrong direction. Because the elasticity solution procedure uses the complete set of own- and cross-price responses, it can move the price in the appropriate direction. Walrasian procedures implicitly have own-price multipliers (usually equal across commodities) and cross-price multipliers (set at zero).

The method by which the relative price change vector is found has some similarity to Newton's method. For instance, Holbrook applies Newton's method to solve a large nonlinear stochastic optimal control problem (5). However, Newton's method uses partial derivatives, not elasticities. Newton's method without damping has been applied to solution of the IIASA and GOL models. It usually converges, but less rapidly than the elasticity solution procedure.

Saari discusses the information required for an iterative solution mechanism which could find equilibrium prices for a pure exchange economy (12). A mechanism which can start with any initial price vector and find an equilibrium price vector has a severe information requirement. Saari finds that the mechanism would need more than the aggregate excess demand function at the current price vector. Even the addition of the matrix of first partial derivatives of excess demand with respect to prices and any higher-order matrices of partial derivatives would not be sufficient to guarantee solution.

Elasticities have three desirable properties for economic applications. First, they help avoid negative prices. They also fit many economic functions well. Third, assumption of constant elasticities implies an exponential model. This means that second and higher-order derivatives are implicitly defined. This specification (if approximately accurate) helps meet some of the information requirements for global solution.

Demand Simultaneous with Prices

In the first type of model to be discussed, demand but not supply is calculated simultaneously with prices. The IIASA world agriculture model fits in this category; in it, supply is calculated prior to demand and is based on the previous year's prices. After the demand elasticity matrix is calculated, solution of this model type involves use of equations 7, 8, and 20. The elasticity of demand is defined here to be the relative change in demand over the relative change in price (equation 3). Equation 3 underlies formulas relating trade and price changes, whereas equation 2 defines the actual numerically estimated elasticities. The total change in demand for a commodity is the sum of the demand changes induced by each price change (equation 4). The new symbols are defined here:

- j = Index for prices or columns.
- N = Number of commodities in system.
- D_i = Demand for commodity_i.
- ΔD_i = Change in demand for commodity_i, total.
- ΔD_{ij} = Change in demand for commodity_i caused by change in price_j.

P_j = World price of commodity j .
 ΔP_j = Change in world price of commodity j .

$$ED_{ij} = \frac{\Delta D_{ij}/D_i}{\Delta P_j/P_j} \quad \text{for all } i \text{ and } j. \quad (3)$$

$$\Delta D_i = \sum_j (\Delta D_{ij}) \quad \text{for all } i. \quad (4)$$

Equations 3 and 4 for each price and a given commodity are combined to give the total relative change in demand (equation 5). An equation 5 for each commodity is combined in matrix form so that the number of equations equals the number of unknowns. The relative demand change vector equals the elasticity matrix times the relative price change vector (equation 6). This system of simultaneous equations may be solved for the relative price change vector.

$$\Delta D_i/D_i = \sum_j (ED_{ij} * \Delta P_j/P_j) \quad \text{for all } i. \quad (5)$$

$$\begin{bmatrix} \Delta D_i/D_i \end{bmatrix}_{Nx1} = \begin{bmatrix} ED_{ij} \end{bmatrix}_{NxN} * \begin{bmatrix} \Delta P_j/P_j \end{bmatrix}_{Nx1} \quad (6)$$

The relative price change vector equals the elasticity matrix inverse times the relative demand change vector (equation 7). World net trade for each commodity is to be driven to zero, so the desired change in demand is the negative of current net imports (equation 8). Forcing world net trade toward zero is equivalent to requiring world exports to equal world imports for a given commodity.

$$\begin{bmatrix} \Delta P_j/P_j \end{bmatrix}_{Nx1} = \begin{bmatrix} ED_{ij} \end{bmatrix}_{NxN}^{-1} * \begin{bmatrix} \Delta D_i/D_i \end{bmatrix}_{Nx1} \quad (7)$$

$$\Delta D_i = (\text{minus}) \text{ net imports}_i \quad \text{for all } i. \quad (8)$$

Assuming constant own- and cross-price elasticities implies that quantities are exponential functions of prices, as is briefly shown below. Integration of equation 5 leads to equation 9, where C_i is a constant of integration and \log_e is the natural logarithm. Exponentiation of equation 9 leads to equation 10, where K_i is a constant and Π denotes the product of the following terms. When K_i is positive the exponential form ensures positive demand, which is an important advantage for economic applications. The same holds true for exponential supply functions. Equation 9 can also form the basis for a solution procedure. The procedure in this report, based on equation 5, converges faster and illustrates the widely applicable equation 20.

$$\log_e(D_i) = C_i + \sum_j [ED_{ij} * \log_e(P_j)] \quad \text{for all } i. \quad (9)$$

$$D_i = K_i * \Pi_j (P_j^{ED_{ij}}) \quad \text{for all } i. \quad (10)$$

Supply and Demand Simultaneous with Prices

For a system in which prices are simultaneous with both demand and supply, the solution for the next iteration is derived below. This solution method is appropriate for reconciling forecasts made by a group of commodity and country analysts. Exact use of this approach requires that the matrix inverse be calculated for each iteration because current levels of supply and demand are used. After the supply and demand elasticity matrices are calculated, solution of this model type involves use of equations 15, 16, and 20. Additional quantity definitions are given here:

- S_i = Supply of commodity i .
- ΔS_i = Change in supply of commodity i , total.
- ΔS_{ij} = Change in supply of commodity i caused by change in price of commodity j .
- ES_{ij} = Elasticity of supply of commodity i with respect to the world price of commodity j .
- NX_i = Net exports of commodity i .
- ΔNX_i = Change in net exports of commodity i .

The elasticity of supply is defined in equation 11. It is similar to the demand elasticity defined in equation 3. The total change in supply for a commodity is the sum of the supply changes induced by each price change (equation 12). The change in net exports is the change in supply minus the change in demand (equation 13). Reordering of equations 3, 4, 11, and 12 allows substitution into equation 13, yielding the change in net exports (equation 14).

$$ES_{ij} = \frac{\Delta S_{ij}/S_i}{\Delta P_j/P_j} \quad \text{for all } i \text{ and } j. \quad (11)$$

$$\Delta S_i = \sum_j (\Delta S_{ij}) \quad \text{for all } i. \quad (12)$$

$$\Delta NX_i = \Delta S_i - \Delta D_i \quad \text{for all } i. \quad (13)$$

$$\Delta NX_i = \sum_j [(S_i * ES_{ij} - D_i * ED_{ij}) * \Delta P_j/P_j] \quad \text{for all } i. \quad (14)$$

Equation 14 may be generalized to matrix notation and solved for the relative price change vector (equation 15). The relative price changes are a function of the net trade changes and the inverse of the quantity-weighted elasticity matrix. The desired change in net exports is the negative of current net exports (equation 16).

$$\begin{bmatrix} \Delta P_j / P_j \end{bmatrix}_{N \times 1} = \begin{bmatrix} S_i * ES_{ij} - D_i * ED_{ij} \end{bmatrix}_{N \times N}^{-1} * \begin{bmatrix} \Delta NX_i \end{bmatrix}_{N \times 1} \quad (15)$$

$$\Delta NX_i = (\text{minus}) \text{ net exports}_i \quad \text{for all } i. \quad (16)$$

Supply and Demand Simultaneous with Prices but Inversion Limited

For some systems in which price is simultaneous with both demand and supply, matrix inversion for every price iteration may be impractical. Other heuristic approaches may be used instead of equation 15. One involves modification of the approach seen earlier for models in which only demand is simultaneous with prices. In this case, demand is changed to the mean of supply and demand. This approach has been used for the GOL model. After the elasticity of quantity is derived (equation 17), solution involves use of equations 16, 19, and 20. This is perhaps the most stable form where the matrix calculation and inversion are performed only once each several years of simulation. It has the advantage that the elements of the resulting "elasticity" matrix may be interpreted approximately as conventional elasticities. These numbers may be used in examining the stability of the system or its aggregate price responsiveness. One new variable is defined below:

EQ_{ij} = Elasticity of quantity of commodity_i with respect to price of commodity_j.

The elasticity of quantity has a special definition here, which is suited to model solution and tuning (equation 17). In the vicinity of equilibrium prices, S_i approaches D_i so that EQ_{ij} approaches ES_{ij} minus ED_{ij} . Thus, EQ_{ij} indicates the aggregate effect on net exports of supply and demand price responsiveness. Equations 4, 12, 13, and 17 may be combined to yield equation 18. Equation 18 is then generalized to matrix notation and solved for the relative price change (equation 19). The relative price changes come from the elasticity matrix inverse and the relative export change vector.

$$EQ_{ij} = \frac{(\Delta S_{ij} - \Delta D_{ij}) / (0.5 * S_i + 0.5 * D_i)}{\Delta P_j / P_j} \quad \text{for all } i \text{ and } j. \quad (17)$$

$$\frac{\Delta NX_i}{0.5 * S_i + 0.5 * D_i} = \sum_j [EQ_{ij} * (\Delta P_j / P_j)] \quad \text{for all } i. \quad (18)$$

$$\begin{bmatrix} \Delta P_j / P_j \end{bmatrix}_{N \times 1} = \begin{bmatrix} EQ_{ij} \end{bmatrix}_{N \times N}^{-1} * \begin{bmatrix} \Delta NX_i / (0.5 * S_i + 0.5 * D_i) \end{bmatrix}_{N \times 1} \quad (19)$$

NEW PRICES

Each new price is the old price as adjusted by the relative price change (equation 20). Treating positive and negative price changes as in equation 20 smoothly and continuously prevents negative prices. The function in equation 20 is smooth in the sense that the function and its first derivative are continuous. The first derivative of the new price with respect to a positive relative price change is given in the upper part of equation 21. The first derivative of the new price with respect to a negative relative price change is shown in the lower part of equation 21. As the relative price change approaches zero, the first derivative of the new price with respect to the negative relative price change approaches $P_{j\text{old}}$. Therefore, the first derivative is continuous.

$$P_{j\text{new}} = \begin{cases} P_{j\text{old}} * (1 + \Delta P_j / P_j) & \text{if } \Delta P_j / P_j \geq 0 \\ P_{j\text{old}} / (1 - \Delta P_j / P_j) & \text{if } \Delta P_j / P_j < 0 \end{cases} \quad \text{for all } j. \quad (20)$$

$$\frac{dP_{j\text{new}}}{d(\Delta P_j / P_j)} = \begin{cases} P_{j\text{old}} & \text{if } \Delta P_j / P_j \geq 0 \\ P_{j\text{old}} * (1 - \Delta P_j / P_j)^{-2} & \text{if } \Delta P_j / P_j < 0 \end{cases} \quad \text{for all } j. \quad (21)$$

Because $P_{j\text{old}}$ is positive, the first derivative is always positive. A solution procedure could instead use a minimum price for each commodity to prevent negative prices. If a price were at its lower bound, its derivative with respect to a desired negative price change would be zero. A more negative desired price change would make no difference. This lack of a price response could slow convergence for a solution procedure which used price bounds.

The distinction between the positive and negative forms in equation 20 is analogous to the difference between Laspeyres and Paasche price indices (14). The nonnegative form uses the old price as the base, whereas the negative form uses the new price. Equation 20 may be derived from equation 22. Equation 22 is reminiscent of Lerner's arc elasticity measure (8). He suggests that the price should be the lesser of the prices at the end points of the arc. Similarly, for the quantity he uses the lesser of the quantities at the end points of the arc. The motivation for his measure is the wish to define an arc elasticity which has the same value when either end point of the arc is taken as the base. Scaling of the prices may be used to keep the numéraire price constant (equation 23). This equation is successful for use with the IIASA model because the model responds to the ratio of each commodity price to the numéraire. The sum of world prices may instead be held constant (equation 24). The latter approach is consistent with the solution mechanisms created by IIASA. The GOL model requires absolute prices, without scaling.

$$\Delta P_j / P_j = (P_{j\text{new}} - P_{j\text{old}}) / \min(P_{j\text{new}}, P_{j\text{old}}) \quad \text{for all } j. \quad (22)$$

$$P_{j\text{scaled}} = P_{j\text{new}} * P_{\text{numéraire old}} / P_{\text{numéraire new}} \quad \text{for all } j. \quad (23)$$


$$P_{j\text{scaled}} = P_{j\text{new}} * \sum_j (P_{j\text{old}}) / \sum_j (P_{j\text{new}}) \quad \text{for all } j. \quad (24)$$

RECALCULATION OF ELASTICITIES

The speed of convergence of the elasticity solution procedure for the March 1985 version of the IIASA model is described in figure 1. It shows the common logarithm of the convergence criterion versus the number of price iterations for each of 19 years. Each line represents the rate of solution for one simulation year. Individual years are not labelled because the feature to be illustrated is the typical range of slopes of the lines. The convergence criterion is the largest ratio of unexplained trade to supply for any commodity. The convergence target for this simulation is 0.00001. The first iteration for each year uses the previous year's equilibrium prices. The figure gives an impression of the rapid decline in unexplained net trade. It also suggests that the logarithm of the convergence criterion is the appropriate scale for describing the rate of convergence with the elasticity solution procedure.

The elasticity solution algorithm can be programmed to require re-estimation of the elasticity matrix whenever the convergence criterion deteriorates within a given year. An increase in the convergence criterion from one price

Figure 1--Convergence rates for 19-year simulation with IIASA model and elasticity solution procedure



Source: Elasticity solution procedure and IIASA world agriculture model.

iteration to the next may be used as the signal requiring a new matrix. This precautionary measure has not been found necessary for solving the IIASA and GOL models.

Based on figure 1, equation 25 has been chosen to specify how frequently the elasticity matrix is to be recalculated if convergence is not yet attained within a given year. The choice of coefficients in this formula is a matter of judgment. Testing of different coefficients in this equation suggests that it is better to err on the side of too many iterations rather than too few before re-estimation of the elasticities. The variables are defined below:

ITERMX = The maximum number of new price vectors which will be tried within a given year before the elasticity matrix will be re-estimated.

FNPREC = Target level of convergence criterion.

$$\text{ITERMX} = \text{largest integer less than } [-0.99 - 2.0 * \log_{10}(\text{FNPREC})] \quad (25)$$

SIMULATION RESULTS

The March 1985 version of the IIASA world agriculture model has been solved for 19 years of endogenous prices (1982 through 2000). Table 1 shows mean numbers of world price iterations to solve the model with different solution procedures. The first column of results comes from running the model with the original solution procedure from IIASA. This procedure is referred to in brief as a Walrasian *tâtonnement* algorithm. If the Walrasian approach does not converge, the procedure switches to use of the Ostrowsky theorem on local convergence and the Kantorovitch lemma on one-sided elimination of infeasibility. Some background for these topics is provided by Marcus and Minc (10). IIASA's solution algorithm requires that the sum of world prices remain constant.

The second algorithm is Newton's method without damping of price changes. The numéraire price is held constant. Equation 25 limits the number of price iterations allowed before re-estimation of the matrix of partial derivatives. During one of the years, Newton's method does not converge to the tightest tolerance level after 101 iterations. For the sake of the comparison, Newton's method is permitted to continue to later years.

The third algorithm is the elasticity solution procedure with the price of the numéraire nonagriculture commodity held constant. The Walrasian approach uses about ten times as many price iterations as does the elasticity procedure at tight tolerances. Newton's method uses an average of two-thirds more price iterations than the elasticity procedure. The relative speed of the elasticity procedure would be more evident if the model were less stable. The elasticity procedure converges at similar rates whether prices are scaled to keep the price sum or the numéraire price constant. The elasticity solution procedure is now installed in the current version of the model at IIASA.

Table 2 lists average numbers of iterations to solve the May 1984 version of the IIASA system. This model has heretofore used a gradient search approach to solve for world prices (6). As with the previously mentioned version of the model, the elasticity solution procedure gives about an order of magnitude improvement over gradient search in the number of iterations required to solve for world prices. Newton's method needs an average of 28 percent more price iterations than the elasticity procedure.

Table 1--Mean number of world price iterations per year to solve March 1985 version of IIASA system for 19 years

Algorithm			
Convergence	:		
Criterion =	:	IIASA -- Walrasian	: Newton's
$\max (NX_i /S_i)$:	tâtonnement;	: method;
over all i	:	constant	: constant
	:	price sum	: numéraire price
	:		
0.01	:	3.4	3.9
0.001	:	18.8	4.6
0.0001	:	54.2	6.8
0.00001	:	113.2	16.1 <u>1/</u>
	:		

1/ Newton's method does not converge in one of the years.

Source: Elasticity solution procedure and IIASA world agriculture model.

Table 2--Mean number of world price iterations per year to solve May 1984 version of IIASA system for 19 years

Algorithm			
Convergence	:		
Criterion =	:	IIASA --	: Newton's
$\max (NX_i /S_i)$:	gradient search;	: method;
over all i	:	constant	: constant
	:	price sum	: numéraire price
	:		
0.01	:	7.4	3.1
0.001	:	23.2	4.3
0.0001	:	77.5	7.7
0.00001	:	136.1 <u>1/</u>	10.4
	:		

1/ The gradient search technique does not converge in two of the years.

Source: Elasticity solution procedure and IIASA world agriculture model.

The convergence criterion is the largest of the net trade-to-supply ratios for all commodities. The version of the IIASA world agriculture model at the Economic Research Service is applied to show absolute and relative changes between policy scenarios and a base run. To show changes which reflect economics and not random differences caused by lack of convergence, the tolerance level 0.00001 is used. There are several reasons for this tight convergence level. Both the base and the scenario runs have errors. There is some random divergence between the runs over 19 simulation years. The model is not highly price-responsive overall; therefore, prices tend to change more than aggregate quantities on a percentage basis. Some countries and commodities are affected much more by world price changes than others. The quantity shocks of policy interest in the Economic Research Service may be small with respect to world supply, yet the results for price-responsive countries may be important. A tight solution tolerance is therefore desirable, and can be attained with a reasonable cost in computer time.

The elasticity solution procedure has been applied to the spreadsheet version of the GOL model, where it enables a tenfold increase in solution speed over Walrasian tâtonnement. In this model, demand and some supply quantities are simultaneous with prices. Because of the nature of the spreadsheet used for the GOL, it is not practical to invert the elasticity matrix with each price iteration. The base quantity for each elasticity is taken to be the average of aggregate supply and demand for the commodity whose net trade change is being measured. The elasticity matrix has also provided part of the information needed for stabilizing and tuning the model (2,7).

REFERENCES

- (1) Brandsma, Andries S., A. J. Hughes Hallett, and Nico van der Windt. "Optimal Control of Large Nonlinear Models: An Efficient Method of Policy Search Applied to the Dutch Economy," Jour. Policy Modeling, 5(2), June 1983, pp. 253-270.
- (2) Conlisk, John. "Quick Stability Checks and Matrix Norms," Economica, 40(160), Nov. 1973, pp. 402-409.
- (3) Fischer, Günther, and Klaus Frohberg. "The Basic Linked System of the Food and Agriculture Program at IIASA: An Overview of the Structure of the National Models," Mathematical Modelling, 3(5), 1982, pp. 453-466.
- (4) Goldfeld, Stephen M., and Richard E. Quandt. Nonlinear Methods in Econometrics. Contributions to Economic Analysis, Number 77. Amsterdam: North-Holland Publishing Company, 1972.
- (5) Holbrook, Robert S. "A Practical Method for Controlling a Large Nonlinear Stochastic System," Annals of Economic and Social Measurement, 3(1), Jan. 1974, pp. 155-175.
- (6) Keyzer, M. A., C. Lemaréchal, and R. Mifflin. Computing Economic Equilibria through Nonsmooth Optimization. Research Memorandum RM-78-13. Laxenburg, Austria: International Institute for Applied Systems Analysis, Mar. 1978.
- (7) Labys, Walter C. Dynamic Commodity Models: Specification, Estimation, and Simulation. Lexington, Mass.: D.C. Heath and Company, 1973.
- (8) Lerner, A. P. "The Diagrammatical Representation of Elasticity of Demand," Review of Economic Studies, 1(1), Oct. 1933, pp. 39-44.
- (9) Liu, Karen, and Vernon O. Roningen. The World Grain-Oilseeds-Livestock (GOL) Model, a Simplified Version. ERS Staff Report No. AGES850128. U.S. Dept. Agr., Econ. Res. Serv., Feb. 1985.
- (10) Marcus, Marvin, and Henryk Minc. A Survey of Matrix Theory and Matrix Inequalities. Boston: Allyn and Bacon, Inc., 1964.
- (11) Roningen, Vernon O., John Wainio, and Karen Liu. The World Grain, Oilseeds, and Livestock Model -- A Microcomputer Version. ERS Staff Report No. AGES850826. U.S. Dept. Agr., Econ. Res. Serv., Sept. 1985.
- (12) Saari, Donald G. "Iterative Price Mechanisms," Econometrica, 53(5), Sept. 1985, pp. 1117-1131.
- (13) Seeley, Ralph. Price Elasticities from the IIASA World Agriculture Model. ERS Staff Report No. AGES850418. U.S. Dept. Agr., Econ. Res. Serv., May 1985.
- (14) Tomek, William G., and Kenneth L. Robinson. Agricultural Product Prices. Ithaca: Cornell University Press, 1981.


APPENDIX

The appendix explains the elasticity solution procedure with flowcharts and a computer program listing. The flowcharts give a visual impression of the sequence of steps in the procedure and suggest their simplicity. The flowcharts are divided into two sections. The first is the main portion of a net trade model. This portion directly runs the price-exogenous (supply) side of the model and updates parameters for the next year. In the price-endogenous years, it then calls the elasticity solution procedure. When the solution procedure has found equilibrium prices, it returns control to the main portion of the model. The model then prepares for the following year.


The second flowchart shows the sequence of steps in the elasticity solution procedure itself. The body of this report elaborates on the details. The procedure selects prices and calls the price-endogenous portion of the model (usually demand). These calls are not indicated by separate flowchart boxes so that the diagram can fit on one page.

The computer program listing shows the actual implementation of the elasticity solution procedure for the IIASA model. The computer listing corresponds to the second flowchart. That is, the solution subroutine is called by the main portion of the IIASA model. In turn, the solution subroutine selects new price vectors and tries them on the price-endogenous portion of the IIASA model. When a satisfactory price vector is found, control returns to the main portion of the IIASA model.

Flowcharts for Elasticity Solution Procedure



Flowcharts for Elasticity Solution Procedure -- Continued



FORTRAN Listing of Elasticity Solution Procedure for IIASA Model

C CREATED BY RALPH SEELEY 8-18-85.
C
C ELASTICITY SOLUTION PROCEDURE FOR MARCH 1985 VERSION OF IIASA MODEL.
C DETERMINE PRICE VECTOR WHICH WILL DRIVE WORLD NET TRADE VECTOR
C TO APPROXIMATELY ZERO. USE NUMERICAL APPROACH TO CALCULATE
C THE ARC ELASTICITY MATRIX AT AN INITIAL POINT.
C THE DESIRED CHANGES IN DEMAND ARE (MINUS) THE NET IMPORT LEVELS.
C SOLVE SYSTEM OF SIMULTANEOUS EQUATIONS FOR RELATIVE CHANGES IN
C PRICES, USED TO CALCULATE NEW PRICES.
C IF EQUILIBRIUM NOT FOUND QUICKLY, ESTIMATE NEW ELASTICITY MATRIX.
C
C ALGEBRAIC REPRESENTATION:
C DI = DEMAND FOR COMMODITY I.
C DDI = CHANGE IN DEMAND FOR COMMODITY I.
C DPJ = CHANGE IN PRICE OF COMMODITY J.
C EDIJ = ELASTICITY OF DEMAND FOR COMMODITY I WITH RESPECT TO THE
C PRICE OF COMMODITY J.
C N = ORDER OF ELASTICITY MATRIX = NUMBER OF COMMODITIES.
C PJ = PRICE OF COMMODITY J.
C
C
$$\begin{bmatrix} DPJ/PJ \\ \vdots \\ DPJ/PJ \end{bmatrix}_{N,1} = \begin{bmatrix} EDIJ \\ \vdots \\ EDIJ \end{bmatrix}_{N,N}^{-1} * \begin{bmatrix} DDI/DI \\ \vdots \\ DDI/DI \end{bmatrix}_{N,1}$$

C
C PJ,NEW = [PJ,OLD * (1 + DPJ/PJ) WHEN DPJ/PJ NONNEGATIVE]
C [PJ,OLD / (1 - DPJ/PJ) WHEN DPJ/PJ NEGATIVE]
C
C FORTRAN VARIABLE DEFINITIONS:
C ELADEM = MATRIX OF DEMAND ELASTICITIES. REPLACED BY ITS INVERSE.
C FN = VALUE OF CONVERGENCE CRITERION, FROM IIASA MODEL, =
C MAX (|NET TRADE I|/SUPPLY I) OVER ALL N COMMODITIES.
C FNPREC = TARGET LEVEL OF CONVERGENCE CRITERION.
C ICOM = POINTER TO MATRIX ROWS (COMMODITIES).
C IONCE = SWITCH TO ENSURE CALCULATION OF ELASTICITIES IN FIRST YEAR.
C ITER = COUNTER: CALLS TO IIASA MODEL SINCE ELASTICITIES ESTIMATED.
C ITERMX = MAXIMUM NUMBER OF PRICE ITERATIONS ALLOWED IN GIVEN YEAR
C BEFORE RECALCULATION OF ELASTICITY MATRIX.
C ITERTL = COUNTER: CALLS TO IIASA MODEL FOR CURRENT SIMULATION YEAR.
C JPRI = POINTER TO MATRIX COLUMNS (PRICES).
C NCOM = N = NUMBER OF COMMODITIES = ORDER OF MATRIX.
C PW = VECTOR OF CURRENT WORLD PRICES.
C PWINIT = VECTOR OF INITIAL WORLD PRICES.
C PWSUMC = CURRENT SUM OF WORLD PRICES.
C PWSUMI = INITIAL SUM OF WORLD PRICES.
C PSHOCK = FRACTIONAL PRICE SHOCK USED TO ESTIMATE ELASTICITY MATRIX.
C RELDEM = VECTOR OF DESIRED RELATIVE CHANGES IN DEMAND.
C RELPRI = VECTOR OF DESIRED RELATIVE CHANGES IN PRICE.
C TY = VECTOR OF WORLD SUPPLY. FROM IIASA MODEL.
C X# = VARIABLES NOT USED BY ELASTICITY SOLUTION PROCEDURE.
C Z = VECTOR OF CURRENT WORLD NET IMPORTS. FROM IIASA MODEL.
C ZINIT = VECTOR OF INITIAL WORLD NET IMPORTS.

SUBROUTINE INTERN (NCOM, IER)

```

COMMON /YEAR/ X1,X2,    FN,    X3,    FNPREC,    X4,X5,X6
COMMON /GEN/  PW(31),   X7,    Z(31),   X8(31,31),X9(31),   TY(31)
DOUBLE PRECISION CWORK, ELADEM, X8, Z
DIMENSION CWORK(10), ELADEM(10,10), PWINIT(10), RELDEM(10),
&           RELPRI(10), X10(32), X11(32), ZINIT(10)
DATA      IONCE /0/,  PSHOCK /0.10/

ITERTL = 1
WRITE (6,*)  ' INTERN: FN',FN, 'ITERTL',ITERTL, 'PSHOCK',PSHOCK

IF (IONCE .NE. 0) GOTO 60
IONCE = 1
C      DETERMINE A REASONABLE MAXIMUM NUMBER OF PRICE ITERATIONS
ITERMX = -0.99 - 2.0 * ALOG10(FNPREC)
WRITE (6,*) ' INTERN: FN PREC',FNPREC, 'ITERMX',ITERMX

C      MAKE NUMERICAL APPROXIMATION OF ARC ELASTICITY MATRIX.
10     CONTINUE

C      STORE INITIAL TRADE, PRICE, AND PRICE SUM.
CPWS    PWSUMI = 0.0
DO 20 JPRI = 1, NCOM
      ZINIT(JPRI) = Z(JPRI)
      PWINIT(JPRI) = PW(JPRI)
CPWS    PWSUMI = PWSUMI + PW(JPRI)
20     CONTINUE

C      SHOCK EACH PRICE, ONE AT A TIME.
DO 40 JPRI = 1, NCOM
      PW(JPRI) = PWINIT(JPRI) * (1.0 + PSHOCK)
C      CALL IIASA MODEL; GIVE PRICE VECTOR AND GET NET TRADE.
      CALL CALCUL (X10,X11,X12)
      ITERTL = ITERTL + 1

C      CALCULATE QUANTITY ARC ELASTICITIES.
DO 30 ICOM = 1, NCOM
      ELADEM(ICOM,JPRI) = (Z(ICOM) - ZINIT(ICOM)) /
      &                      (TY(ICOM) + ZINIT(ICOM)) / PSHOCK
30     CONTINUE
      WRITE (6,*) ' INTERN: PW, Z'
      WRITE (6,90000) (PW(ICOM), ICOM = 1, NCOM),
      &                  (Z(ICOM), ICOM = 1, NCOM)
      PW(JPRI) = PWINIT(JPRI)
40     CONTINUE

C      RESTORE INITIAL NET IMPORTS WHICH CORRESPOND TO INITIAL PRICES.
      WRITE (6,*)  ' INTERN: ELADEM'
      DO 50 ICOM = 1, NCOM
          Z(ICOM) = ZINIT(ICOM)
          WRITE (6,90000) (ELADEM(ICOM,JPRI), JPRI = 1, NCOM)
50     CONTINUE

C      INVERT THE ELASTICITY MATRIX IN PLACE. ANY GENERAL MATRIX
C      INVERSION SUBROUTINE MAY BE USED. FLAG ANY INVERSION ERROR.
      CALL DINVT (ELADEM, NCOM, 10, CWORK, IER)
      WRITE (6,*) ' INTERN: IER', IER

```

```

60      CONTINUE
C      CALCULATE NEW PRICE VECTOR AND FIND NEW NET TRADE VECTOR.
      DO 110 ITER = 1, ITERMX
C      CALCULATE VECTOR OF DESIRED RELATIVE CHANGES IN DEMAND.
      DO 70 ICOM = 1, NCOM
         RELDEM(ICOM) = -Z(ICOM) / (TY(ICOM) + Z(ICOM))
70      CONTINUE

C      CALCULATE VECTOR OF RELATIVE CHANGES IN PRICES.
      DO 80 ICOM = 1, NCOM
         RELPRI(ICOM) = 0.0
         DO 80 JPRI = 1, NCOM
            RELPRI(ICOM) = RELPRI(ICOM) +
&               ELADEM(ICOM,JPRI) * RELDEM(JPRI)
80      CONTINUE

C      USE RELATIVE PRICE CHANGES TO FIND NEW PRICES.
CPWS    PWSUMC = 0.0
      DO 90 JPRI = 1, NCOM
         RELPRJ = RELPRI(JPRI)
         IF (RELPRJ .GT. 0.0) PW(JPRI) = PW(JPRI) * (1 + RELPRJ)
         IF (RELPRJ .LT. 0.0) PW(JPRI) = PW(JPRI) / (1 - RELPRJ)
CPWS    PWSUMC = PWSUMC + PW(JPRI)
90      CONTINUE

&      WRITE (6,*) ' INTERN: OLD Z, OLD TY; NEW RELDEM, ',
&      'NEW RELPRI, NEW PW BEFORE SCALING, NEW PW AFTER SCALING'
&      WRITE (6,90000) (Z(ICOM),ICOM=1,NCOM), (TY(ICOM),ICOM=1,NCOM),
&      RELDEM, RELPRI, (PW(ICOM), ICOM = 1, NCOM)

C      SCALE PRICES TO HOLD CONSTANT THE NUMERAIRE PRICE OR PRICE SUM.
      DO 100 JPRI = 1, NCOM
         PW(JPRI) = PW(JPRI) * PWINIT(NCOM) / PW(NCOM)
CPWS    PW(JPRI) = PW(JPRI) * PWSUMI      / PWSUMC
100     CONTINUE
      WRITE (6,90000) (PW(ICOM), ICOM = 1, NCOM)

C      MAKE FUNCTION EVALUATION AND CHECK FOR CONVERGENCE.
C      CALL IIASA MODEL; GIVE PRICE VECTOR AND GET NET TRADE VECTOR.
      CALL CALCUL (X10,X11,X12)
      ITERTL = ITERTL + 1
      WRITE (6,*) ' INTERN: FN',FN, 'ITERTL',ITERTL
      IF (FN .LT. FNPREC) RETURN
      IF (ITERTL .GT. 100) STOP ' INTERN: ITERTL OVER 100'
110     CONTINUE

      GOTO 10
90000 FORMAT (10(1X,G12.6))
      END

```

NATIONAL AGRICULTURAL LIBRARY



1022384317

Ar



1022384317